

HP 82929A Programmable ROM Drawer Instruction Sheet

The HP 82929A Programmable ROM Drawer allows you to incorporate your own software into the HP Series 80 computer system. PROMs (programmable read-only-memory units) that you have *already* programmed (using an assembler ROM and PROM programmer) fit into the drawer, which has two sockets to accommodate either two 4K-byte (2732) EPROMs or one or two 8K-byte (2764) EPROMs (erasable programmable ROMs).

WARNING

Do not place fingers, tools, or other foreign objects into the computer plug-in ports. Such actions may result in minor electrical shock hazard and interference with pacemaker devices worn by some persons. Damage to the computer port contacts and internal circuitry may also result.


Installing Programmable ROMs

Programmable ROMs are used with the HP Series 80 computers by first installing them in the PROM drawer and then installing the PROM drawer into one of the ports of the computer.

If you are using 2732 EPROMs (4K bytes each), you must install two, providing 8K bytes of storage. If you are using 2764 EPROMs (8K bytes each), you may install one or two, providing 8K or 16K bytes of storage. In either case, you can install up to *but no more than* three programmable ROM drawers, achieving a maximum of 24K (with all 2732 EPROMs) to 48K (with all 2764 EPROMs) bytes of storage.


Note: Do not install duplicate (P)ROMs in the PROM and/or standard ROM drawers. Duplicate (P) ROMs will create error conditions.

1. Remove the six screws from the underside of the drawer and lift apart the case top and bottom. Turn the printed-circuit board over to expose the integrated circuits.
2. Insert the *programmed* PROM(s) into the U18/U19 socket(s) after aligning the PROM pins with the socket pins. Note that 2732 EPROMs do not use the top four pins of the socket (see the markings on the board).



For 2732 EPROMs: the PROM with the *first* 4K bytes (0K to 3K) must be placed into the socket marked U18; the PROM with the *second* 4K bytes (4K to 7K) must be placed into socket U19.

For 2764 EPROMs: it does not matter which socket receives which PROM.



If necessary, the PROMs can be removed by carefully prying the PROM out of the socket with a small, flat-blade screwdriver.

3. Insert the jumper labeled **W1** (see photo) into the slot with a number corresponding to the EPROM type (2732 or 2764) being used. (The factory jumper setting is for 2732.)
4. Set the switches labeled **S1** and **S2** to the proper bank addresses by moving the toggles to be counted to 1. (It is easiest to use a pencil or other fine-tipped instrument for this.) All other toggles should be set to zero. **Note the printing on the printed-circuit board indicating which toggle position is 0 (or "ON") and which is 1.**



There are 256 (2^8) possible bank addresses ("0" to "255"). The diagram below, right shows the decimal value of each toggle. The values of the toggles are *additive* (refer to the table of example settings), so if all toggles are set to 1, you obtain the maximum address value of 255.

- A setting of all ones (1s) will effectively turn that switch's logic off since the system does not check for that address.
- A setting of all zeros (0s) will cause faulty PROM operation, since that is the system's ROM address.

For 2732 EPROMs: program only **S1** and set **S2** to all ones.

For 2764 EPROMs: socket **U18** corresponds to switch **S1** and socket **U19** corresponds to switch **S2**. If only one PROM is being used in the drawer, program only the corresponding switch and set the other switch to all ones. If two PROMs are being used, program both switches. (If **S1** and **S2** are programmed to the same address, **U18** will be selected because **S1** has priority over **S2**.)

5. Replace the printed-circuit board in the PROM drawer. (The board fits upside down over the case section that has the HP label on the outside.) Two of the diagonally positioned screw holes have alignment guides: holes and corresponding sockets that fit onto each other. Match these guides to ensure proper placement of the board. Hold the grounding clamp in place while screwing the case back together.



Example Toggle Settings

Address (Value)	1 (1)	2 (2)	3 (4)	4 (8)	5 (16)	6 (32)	7 (64)	8 (128)
1	1	0	0	0	0	0	0	0
8	0	0	0	1	0	0	0	0
173	1	0	1	1	0	1	0	1
254	0	1	1	1	1	1	1	1

Installing and Removing the Programmable ROM Drawer

CAUTION

Always turn off the computer *before* installing or removing the PROM drawer. If this is not done, the system may be damaged.

The computer is designed to operate with *no more than* three PROM drawers installed at any time. Installing more than three PROM drawers may create error conditions and can damage the system.

To install the PROM drawer:

1. Turn off the computer and any connected peripherals.
2. Remove the protective cover from any unused port. Remaining unused ports should be kept covered.

CAUTION


Do not force the PROM drawer into the port. If the PROM drawer is upside down, forcing it further may result in damage to the computer or the PROM drawer itself.

3. Position the PROM drawer with the HP label facing up, and insert the contact end into the port. Slide it in until the grips meet the sides of the port. A slight side-to-side motion may be necessary to seat the PROM drawer in the port. The tracks are keyed to prevent the PROM drawer from being installed upside down.

The capabilities of the installed PROMs are now available to the HP Series 80 system. You do not have to remove the PROM drawer unless you want to change the PROMs in your system.

To remove the PROM drawer:

1. Turn off the computer and any connected peripherals.
2. Firmly grasp the PROM drawer and pull it from the port. Store the PROM drawer in its original container or where it will be safe from damage.
3. Replace the protective cover port.



Troubleshooting

You can check the functionality of your PROMs and PROM drawer by giving your HP Series 80 computer a command unique to the PROM(s) installed. If the computer responds with an error message, you should check the system:

1. Turn off the computer and all peripherals. After disconnecting the PROM drawer and all other plug-in devices from the ports, turn on the computer. If the cursor appears and no error message is displayed, the computer itself is functioning properly.
2. Disassemble the PROM drawer. Check the jumper setting and the dip switch (S1 and S2) settings. These are common sources of error. Make sure the jumper is set to the EPROM type number being used and the dip switches are set correctly to the proper bank addresses.
3. If all settings are correct, then check the condition of your PROM(s) by replacing the new or questionable PROM(s) with (a) *known* good PROM(s)—making sure the jumper and new address are set correctly. Turn off the computer and re-install the drawer. Turn on the computer and test a PROM command again.
 - If the computer responds correctly and without an error message, then the PROM drawer is functioning properly, but the original PROM was not.
 - If you still get a syntax error message, the PROM drawer is at fault.
4. If either the PROM drawer or the computer (step 1) does not operate properly, repair service is required.

Sample BASIC Program to interface the HP Series 80 Assembler ROM with a *Data I/O System 19* PROM Burner

The following is a sample BASIC program which can be used to transfer your assembler programming code from the HP Series 80 Assembler ROM to a *System 19* PROM burner (from *Data I/O*) via a serial interface. It is included as a reference aid. For other PROM burner systems, you would need to modify lines 360 to 760 according to the system's particular characteristics.

```

10 DIM C$(8192),R$(125)
20 CLEAR
30 DISP "OBJECT CODE FILENAME";
40 INPUT N$
50 ON ERROR GOTO 800
60 ASSIGN# 1 TO N$
70 OFF ERROR
80 ! *****
90 ! * READ IN ROM BYTES FROM THE DATA FILE *
100 ! *****
110 FOR I=1 TO 100000
120 READ# 1 ; R$
130 ! *****
140 ! * WHEN ALL OF THE DATA HAS BEEN READ, THE NEXT STRING WILL BE A NULL STRING. *
150 ! *****
160 IF R$="" THEN 190
170 C$[(I-1)*125+1,(I-1)*125+LEN(R$)]=R$ ! BUILD UP ONE TOTAL STRING
180 NEXT I
190 CLEAR @ DISP "GENERATING CHECKSUM" @ C=0 @ C$(8192)=" "
200 FOR I=1 TO 8189 STEP 2
210 ! *****
220 ! * THE FOLLOWING CODE GENERATES THE CHECKSUM THAT MUST GO IN THE LAST *
230 ! * BYTES OF THE ROM. *
240 ! *****
250 C=C+NUM(C$(I))+NUM(C$(I+1))*256
260 NEXT I
270 C1=2^16*FP(C/2^16)
280 C2=IP(C/2^16)
290 C=ABS(C1+C2-2^16+1)
300 C$(8191)=CHR$(256*FP(C/256))&CHR$(IP(C/256))
310 ! *****
320 ! * THE FOLLOWING LINES OF CODE WILL BE DIFFERENT DEPENDING *
330 ! * UPON THE EPROM PROGRAMMER USED. THIS EXAMPLE USES THE *
340 ! * "DATA I/O SYSTEM 19" PROGRAMMER. *
350 ! *****
360 OUTPUT 10 USING "#,K" ; "FY 19/24"%CHR$(13) ! WE'RE USING 2732's
370 OUTPUT 10 USING "#,K" ; "FM 30"%CHR$(13) ! SET SERIAL TRANSMISSION MODE
380 OUTPUT 10 USING "#,K" ; "DI 0/1000"%CHR$(13) ! SEND BYTE COUNT & START ADDRESS
390 OUTPUT 10 USING "#,A" ; "" ! SEND A START TRANSMISSION BYTE
400 BEEP @ DISP @ DISP "SENDING DATA"
410 ! *****
420 ! * NOW SEND 4K BYTES *
430 ! *****
440 FOR I=1 TO 4096
450 D$=DIO$(NUM(C$(I)))
460 OUTPUT 10 USING "#,AAAA" ; D$(4,6)&" "
470 NEXT I
480 OUTPUT 10 USING "#,A" ; "" ! SEND END OF TRANSMISSION BYTE
490 BEEP
500 CLEAR @ DISP "FIRST HALF SENT"
510 DISP @ DISP "INSERT 2732 EPROM INTO SOCKET, THEN PRESS 'CONT'."
520 PAUSE
530 DISP @ DISP "PROGRAMMING."
540 OUTPUT 10 USING "#,K" ; "" ! SEND A 'BREAK'
550 OUTPUT 10 USING "#, K" ; "PG"%CHR$(13) ! SEND THE PROGRAM COMMAND
560 DISP @ DISP "PRESS 'CONT' WHEN READY TO SEND DATA FOR SECOND HALF."
570 PAUSE
580 OUTPUT 10 USING "#,K" ; "" ! SEND A 'BREAK'
590 OUTPUT 10 USING "#,K" ; "DI 0/1000"%CHR$(13) ! SEND BYTE COUNT AND START ADDRESS
600 OUTPUT 10 USING "#,A" ; "" ! SEND A START OF TRANSMISSION BYTE
610 BEEP @ DISP @ DISP "SENDING SECOND HALF DATA"
620 ! *****
630 ! * SEND 4K BYTES OF DATA *
640 ! *****
650 FOR I=4097 TO 8192
660 D$=DIO$(NUM(C$(I)))
670 OUTPUT 10 USING "#,AAAA" ; D$(4,6)&" "
680 NEXT I
690 OUTPUT 10 USING "#,A" ; "" ! SEND END OF TRANSMISSION BYTE
700 BEEP
710 CLEAR @ DISP "SECOND HALF SENT"
720 DISP @ DISP "INSERT 2732 EPROM INTO SOCKET, THEN PRESS 'CONT'."
730 PAUSE
740 DISP "PROGRAMMING."
750 OUTPUT 10 USING "#,K" ; "" ! SEND A 'BREAK'
760 OUTPUT 10 USING "#, K" ; "PG"%CHR$(13) ! SEND THE PROGRAM COMMAND
770 PAUSE
780 DISP "PROGRAMMING FINISHED."
790 STOP
800 DISP "ERROR ";ERRN @ GOTO 30
810 END

```

Warranty and Repair Service Information

The complete warranty statement and procedures for obtaining repair service are contained on the Warranty and Service sheet shipped with your HP 82929A Programmable ROM Drawer. If you need additional information, please contact your authorized HP dealer or the nearest Hewlett-Packard sales and service facility. The addresses of these facilities are listed at the back of the owner's manual for your HP Series 80 computer.

Place this instruction sheet in your HP Series 80 system binder for future reference.



1000 N.E. Circle Blvd., Corvallis, OR 97330